

1. 計算練習

BASIC の命令	意味
A+B	A+B
A-B	A-B
A*B	A×B
A/B	A÷B
A/B	AをBで割った商
A MOD B	AをBで割った余り

問題1 次の計算をせよ。

- (1) $(5*3)-7$
- (2) $(13\#6)^{(7-3)}+(9 \text{ MOD } 4)$
- (3) $6-7*3^2$

BASIC の命令	意味
ABS(X)	Xの絶対値
INT(X)	X以下で最大の整数
SQR(X)	Xの正の平方根
SGN(X)	X>0 のとき 1, X=0 のとき 0, X<0 のとき -1 となる

例

$ABS(-9.1)=9.1$
 $INT(3.14)=3$ $INT(-3.14)=-4$ $INT(3)=3$
 $SGN(-5.8)=-1$ $SGN(ABS(-5.8))=SGN(5.8)=1$

例

$A*SGN(A)$ $ABS(A)$
 $(SQR(A))^2$ A

- ◇ 計算に使うカッコは小カッコ()だけで, { }や []は使わない。
- ◇ 計算順序は普通と同じ。
 カッコ
 累乗(^)
 *,/,#,MOD
 +,-

- ◇ INT は(少数切り捨て)の意味だが, 負の数では注意。
- ◇ SQR(X) の(X)に負の数を入れてはいけない。

問題2 次の計算をせよ。

- (1) $SGN((-3)^3)$
- (2) $INT((5-3*7)/5)$

問題3 次の BASIC の式を普通の式に直せ。

- (1) $(A-5)^B+(3*A-B)$
- (2) $(A*B*C-D)/(E*F)$

問題4 Aをの $\sqrt{5}$ 小数部分, Bをの $\frac{\sqrt{5}+2}{2}$ 小数部分とするとき, A, Bを BASIC の式で表せ。

問題5 次の式を BASIC の式に直せ。

- (1) \sqrt{AB}
- (2) $A^2X^2+ABX-BC-ACX$

参考 その他の関数 _____

- ◇ $SIN(X*3.14159/180)$... $\sin X$ 度
 $COS(x*3.14159/180)$ や $TAN(X*3.14159/180)$ も同様。
 カッコの中に入る角度は, 本当はラジアンという単位。

$$(q \text{ 度} = \frac{p}{180} q \text{ ラジアン})$$

- ◇ $LOG(S)/LOG(R)$... $\log_R S$

$$\log_R S = \frac{\log_a S}{\log_a R} \quad \text{から。}$$

本当は $LOG(S)$ は $\log_e S$ を求めている。($e=2.718...$)

2.変数とプログラム

変数 数学で使う文字のようなもの。ただし、2文字以上でも1つの変数として扱う。だから $A*B$ と AB は別のもの。

プログラム **例**

```
100 REM ローン返済シミュレーション
110 R=1.05 : S=0
120 INPUT "借入れ額";A
130 INPUT "毎年返済額";B
140 INPUT "返済年数";N
150 FOR Y=1 TO N
160 A=A*R
170 A=A-B : S=S+B
180 IF A<=0 THEN S=S-ABS(A) : GOTO 230
190 NEXT Y
200 PRINT N"年後のローン残高は";A;"円です"
210 PRINT "その間に支払った総額は";S;"円です"
220 END
230 PRINT Y;"年で返済できます。"
240 PRINT "その間に支払った総額は";S;"円です"
250 END
```

実行結果

```
借入れ額? 5000000
毎年返済額? 700000
返済年数? 10
10年で返済できます。
その間に支払った総額は 6339948円です
```

この3つの数字はキーボードから入力する

プログラムの規則

(1) 行番号の小さい順に実行する。

100 110 120 ...

例外 150~190(繰り返し), 220(終了)など

(2) 1行に2つ以上の文章を書くときは:(コロン)をつける。

◇ :と;(セミコロン)は違うことに注意。

代入

$R=1.05$ などが代入であり、この時、 R という変数に 1.05 が代入される。この後、 $20*R$ という命令があったとしたら、それは 20×1.05 を意味する。

◇ $A=B+C$ という場合、 A が $B+C$ に変化する。このように、代入では必ず変化するものが左辺にくる。

◇ 代入文の前に LET と書くことがあるが、99%省略される。

例 LET A=5 A=5

◇ $A=A+5$ は、 A に5を加えることを意味する。

問題6 次のプログラムを実行した後の A の値を求めなさい。

```
10 A=3 : B=5 : C=7
20 A=A-B+C+1 : B=C-A
30 C=C+1 : A=A+B+C
40 END
```

(ヒント) 次のような表を作る。

行	A	B	C
10	3	5	7
20			
30			
40			

この表を、このプリントでは変数表と呼ぶ。

3.PRINT 文(BASIC の超基本)

PRINT 文

PRINT (何か)

で画面に文字や数字を表示する。センター試験では表示される数字や文字が回答の対象であるから、正確に理解する必要がある。

- ☆ 前ページの問題6は、実行しても画面に何も表示されない。結果が知りたいければ、最後に PRINT を使って結果を画面に表示させなくてはならない。

(1) 数や式の場合

BASIC の文章	表示結果
PRINT 3*5-7	8
PRINT ABS(-7.1)	7.1
A=3 : PRINT A	3
A=3 : B=5 : PRINT A+B-1	7

PRINT のあとに数字や変数の式がある場合、その計算結果が表示される。

(2) 文字の場合

BASIC の文章	表示結果
PRINT "HELLO!"	HELLO!
PRINT "THANK YOU"	THANK YOU
PRINT "3*5-7"	3*5-7
A=3 : PRINT "A"	A
A=3 : B=5 : PRINT "A+B-1"	A+B-1

PRINT のあとに文字を表示させるときには、"(ダブルクォーテーション)で表示したい文字を囲う。"で囲われた部分は何であろうともそのまま表示する。

(3) 「つなぎ」の記号

```
10 PRINT "SEE"  
20 PRINT "YOU"  
30 PRINT "AGAIN"  
40 END  
  
10 PRINT "SEE";  
20 PRINT "YOU";  
30 PRINT "AGAIN"  
40 END  
  
10 PRINT "SEE",  
20 PRINT "YOU",  
30 PRINT "AGAIN",  
40 END
```

```
SEE  
YOU  
AGAIN  
  
SEEYOUAGAIN  
  
SEE    YOU    AGAIN
```

のように、PRINT のあとに何も置かないと、その次に表示される文字は行を改めて表示される。

のように、PRINT のあとに、(セミコロン)を入れると、その次に表示される文字は前に PRINT したものに続けて表示される。

のように、PRINT のあとに、(カンマ)を入れると、その次に表示される文字との間にすき間ができる。

(4) 複合

BASIC の文章	表示結果
PRINT "3*5-7=";3*5-7	3*5-7=8
PRINT "ABS(-7.1)とは" ; ABS (-7.1) ; "のこと"	ABS(-7.1)とは 7.1 のこと

- ☆ 2番目の文章は、下のように考えるとよい。どこが"で囲われているのかに注意。

PRINT "ABS(-7.1)とは";ABS (-7.1) ; "のこと"

INPUT 文

INPUT (質問)(変数)
で、変数に数値を入力する。

例

```
10 INPUT "A=";A
20 PRINT A
30 END
```

実行結果

```
A=? 5
5
```

```
A=? 10/5+7
9
```

```
A=? SQR(3)
1.73205
```

◇ プログラムではついていなかった?がついていることに注意。

また、一度に二つ以上の変数を入力できる。

例

```
10 INPUT "A,B,C=";A,B,C
20 PRINT A,B,C
30 END
```

実行結果

```
A,B,C=? 3,21,7
3      21      7
```

問題 7 次の命令を実行すると、画面にどう表示されるか示せ。

(1) PRINT 2-5*SGN(-2)

(2) A=5 : B=8 : PRINT "8*5=";A*8

問題 8 次のプログラムを実行したとき、画面にどう表示されるか示せ。

```
10 A=5 : B=8 : C=3
20 A=(A+C)/2 : PRINT A,
30 B=B-A : PRINT B,
40 C=C^2 : PRINT C,
50 A=(A+B+C)*SGN(A+B+C)
60 PRINT "A=";A : PRINT "B=";B : PRINT "C=";C
70 D=(A+B+C)/3 : PRINT D
80 END
```

問題 9 次のプログラムは、入力した A についての何を計算し、表示するものか答えよ。

```
10 INPUT "A=";A
20 A=A*A*A*A
30 PRINT A
40 END
```

4.FOR ~ TONEXT 文(最重要!!)

FOR ~ TONEXT 文

同じような操作を繰り返すときには、FOR ~ TONEXT を使う。プログラムを理解する上でもっとも重要な構文だから、絶対に理解しないとイケない。

例

```
10 S=0
20 FOR A=1 TO 100
30   S=S+A
40 NEXT A
50 PRINT S
60 END
```

実行結果

5050

FOR ~ TONEXT は、FOR ~ TO の部分(20行)と、NEXT 部分(40行)の二つを組にして使う。

FOR (変数) = (最初の値) TO (終わりの値)

で、まず繰り返しを数える変数と、その最初の値、終わりの値を決める。上の例のプログラムでは、A をはじめに1にして、A が100 になるまで繰り返す。繰り返しは

NEXT (変数)

まで続ける。(変数)にはFOR で使った変数と同じ物が入る。NEXT に到達すると、プログラムはまたFOR のある行まで戻る。この時、繰り返しを数える変数の値が1 増える。例のプログラムを考える。(右側に変数表を記す。)

10 行で、S=0 を代入する。次に、A=1 とし、A が100 になるまで繰り返し作業を行う。30 行では、S に A を足してS が1 になる。40 行でNEXT A によってプログラムはFOR のある20 行へ戻る。その時にA には1 が加えられて2 になる。続いて、再び30 行でS にA が加えられるが、この時、A は2 になっているので、S は1+2 で3 になる。40 行でNEXT A によりプログラムは20 行に戻り、その時にA には1 が加えられて3 になる。この操作をA が100 になるまでとことん繰り返していく。.....とことん繰り返す..... A=99 のとき、S は4950 である。続いて、20 行でA に1 を加えてA=100 にする。FOR ~ TO.....NEXT の構文は、(終わりの値)になったとき(この場合は100)まで繰り返して終わるから、30 行へ行き、S に100 を足してS は5050 になる。この次のNEXT では20 行に戻らず、そのまま次の50 行へ行く。ここでS の値(5050)を表示して、60 行で終了する。

行番号	A	S
10		0
20	1	
30		1
40	2	
20		3
30	3	
40		6
20		
30		
40		
20	99	
30		4950
40		
20	100	
30		5050
40		
50		
60		

END 文

プログラムの途中で END があれば、プログラムは次の行へ行かないでそこで終了する。普通、プログラムは最後の行で終了するが、途中で終わらせたいときには END を使う。

STEP 命令

FOR (変数) = (最初の値) TO (終わりの値) STEP (増加量)
で、NEXT から FOR に戻るときに変数が増える量を決める。STEP
を省略すると、NEXT から FOR に戻るときに 1 ずつ増加する。

例

```
10 S=0
20 FOR A=1 TO 51 STEP 2
30   S=S+A
40 NEXT A
50 PRINT S
60 END
```

実行結果

676

A は最初 1 であり 30 行で S に 1 が加えられて S は 1 になる。次に 40 行の NEXT で 20 行に戻るとき、A は 2 増加して 3 になり、30 行で S には 3 が加えられ S は 4 になる。つまり S には $1+3+5+\dots+51$ と、1 から 51 までの奇数が加えられるのだから、

$$1+3+5+\dots+(2n-1) = 2^n \left[\begin{array}{l} \text{数学の「個数の処理」, または} \\ \text{数学 A の「数列」参照} \end{array} \right]$$

の公式より、 $1+3+5+\dots+(2 \cdot 26-1) = 26 \times 26 = 676$ であり、画面には 676 が表示される。

☆ FOR 文と NEXT 文とで囲まれた部分は、見やすくするためにほかの行より少し下げるのが一般的である。

入れ子構造 (ネスト)

FOR ~ TO.....NEXT の繰り返しの中に、また FOR ~ TO.....NEXT を入れることができる。これを入れ子構造 (ネスト) という。

例

```
10 FOR A=1 TO 5
20   FOR B=1 TO A
30     PRINT A;
40   NEXT B
50   PRINT
60 NEXT A
70 END
```

実行結果

1
22
333
4444
55555

20 ~ 40 のループでは A に代入された数値を画面に A 個表示する。(; にも注意!) これを A が 1 から 5 まで繰り返す。ここで、50 行の PRINT に注意する。50 行では何も表示しないように思われるが、PRINT 文の解説のところにあったように、PRINT の最後に何も無いときには、次に行を変えて表示する。だから、この PRINT は「次に表示するものは、行を変えて表示する」ということを示している。A による 20 行から 40 行のループが 1 回終わるたびに、次の数字は行を変えることになる。

☆ 入れ子構造は幾重にも作ることができるが、FOR 文と NEXT 文はカッコの対応のように、順番に始まり、順番に終わらなくてはならない。(これはプログラムを書く時の話だから、センター試験にはあまり重要ではない。)

悪い入れ子構造の例

```

10 FOR A=2 TO 20 STEP 2
20   FOR B=5 TO 500 STEP 5
30     .....
40     .....
50     .....
60     .....
70   NEXT A
80 NEXT B
90 END

```

これでは、Bのループが終わる前にAのループが終わってしま
ってわけが分からない。

「彼は『こんにちは』とあいさつした」という文章のようなも
ので、これを正しくするには、

```

10 FOR A=2 TO 20 STEP 2
20   FOR B=5 TO 500 STEP 5
30     .....
40     .....
50     .....
60     .....
70   NEXT B
80 NEXT A
90 END

```

のようにすればよい。

5. IF ~ THEN 文(条件)

IF ~ THEN 文 (もし~ならば, そのときは...)

IF (条件) THEN (命令)

条件が満たされているときに, THEN 以下を実行する。

条件が満たされていないならば, 次の行に進む。

条件に使われる記号

BASIC の記号	意味
A>B	A>B
A<B	A<B
A>=B または A=>B	A B
A<=B または A=<B	A B
A<>B または A><B	A b

例 IF A=B THEN A=A*2

(A と B とが等しければ, A を 2 倍する)

IF A*B>=100 THEN GOTO 100

(A×B が 100 以上ならば, 100 行へ行く。)

◇ THEN の後の GOTO は略してよい。

IF A*B>=100 THEN 100

IF A<>B THEN A=30:GOTO 200

(A と B とが等しくなければ, A を 30 にして 200 行へ行く)

◇ A と B とが等しい時には, すぐに次の行へ行って, 200 行へ
は行かない。

GOTO

GOTO (行番号) でその行番号へ行く。

例 GOTO 30 で 30 行へ行く。

6.GOSUB 文,RETURN 文

GOSUB 文, RETURN 文

GOSUB は GOTO に似て,指定された行番号へ行くが,RETURN があつたときに GOSUB があつた行の次の行に戻ってくる。

例

```
10 FOR A=1 TO 7
20   GOSUB 100
30   PRINT S
40 NEXT A
50 A=10 : GOSUB 100
60 PRINT S
70 END
100 S=1
110 FOR B=1 TO A
120   S=S*B
130 NEXT B
140 RETURN
```

20 行で 100 行へ行く。110 行から 130 行のループでは S に 1 からまでの数をどんどんかけていく。つまり,このループでは A の階乗 (A!と書く。A!=1×2×3×...×A) を求めている。このループが終わったら,140 行の RETURN に行く。ここで,このループには 20 行からやってきたから,その次の行の 30 行へ戻る。30 行では 100 行以下で計算した A!の値を表示する。これを A が 1 から 7 まで繰り返す。10 行~40 行のループが終わると,50 行では A に 10 を代入して 100 行へ行く。ここでは同様に A!,つまり 10 の階乗の値を求める。そして,

140 行の RETURN で戻るのだが,今回は 50 行からやってきたので,50 行の次の 60 行へ戻る。60 行では 30 行と同様に 10 の階乗を表示する。そして 70 行で終了する。(END 文の説明参照)

よって,画面には 1 から 7 までの階乗と,10 の階乗が表示される。

実行結果

```
1
2
6
24
120
720
5040
3628800
```

問題 10 次のプログラムを実行したときに,プログラムはどういう順番で処理をするか。行番号の順番を答えよ。

(例 10 20 30 40)

```
10 GOTO 70
```

```
20 REM 最初に REM とおくと,その後の文章は
```

```
30 REM 何の意味もなくなります。
```

```
40 ' この記号でも同じ。
```

```
50 RETURN
```

```
60 GOTO 130
```

```
70 FOR A=1 TO 3
```

```
80   REM この命令はプログラムに説明を
```

```
90   ' つけたいときに使います。
```

```
100 NEXT A
```

```
110 GOSUB 20
```

```
120 GOTO 60
```

```
130 GOSUB 30
```

```
140 END
```


7.その他

RND

RND(1)で、0以上1未満の数をつつ、ランダムに発生させる。

BASIC の命令	実行結果 (例)
PRINT RND(1)	0.5689142

例 A=INT(RND(1)*6)+1

RND(1)は0以上1未満の数であるから、これを6倍すると、0以上6未満の数になる。更に、INTでこの小数点以下を切り捨てるから、INT(RND(1)*6)は0,1,2,3,4,5のどれかになる。更にこれに1を加えるから、Aは1,2,3,4,5,6のどれかになる。つまり、これでさいころの代わりができる。

◇ 但し、実際にはRANDOMIZEという命令を始めに使う必要がある。(数学Bの教科書を参照)

問題11 次のプログラムを実行したときに、YESと表示される確率を求めよ。

```
10 A=INT(RND(1)*6)+1
20 IF A<=2 THEN PRINT "YES": END
30 PRINT "NO": END
```

AND, OR, NOT

IF文の条件部分に、AND, OR, NOTという文字が出てくることがある。これは、それぞれ数学でいう「かつ()、または()、……でない()」を表す。

例 A>=5 AND A<10 は、Aが5以上かつBが10未満であることを表す。

A>=100 OR NOT B MOD 2=0 は、Aが100以上であるか、またはBを2で割った余りが0でない(つまり奇数)ことを表す。

8.問題の解き方

センター試験コンピュータの問題は、次のような流れのものが基本である。

INPUTで数値を入力する

それに対して文字や数字が表示される。

これに対しては、次のようにして解いていく。

最初の設問は具体的な数字に関する計算問題であることが多いので、丁寧に変数表を書いて計算する。最初に具体的な数字が出てこない時は、自分でA=5だったら、などを考えてみる。

何回か具体例をやることで、このプログラムが「何を求め、どう表示する」ものかを理解する。その上で、プログラムどおりでなくても自分でもっと早く計算する方法を編み出す。(例 1からNまでの和を求めるのであれば $\frac{1}{2}n(n+1)$ など。そうすれば、Nに100を入力するとか、1から100までのすべての数を入力するなどの問題にも対処できる。むしろ、これが出来ないと面倒くさい計算をたくさんして、数列のほうがよかった、ということになってしまう。

わからない文が出てきたり、プログラムでないもの(可能性はある)が出てきたら、迷わず数列やベクトルなど、他のものに逃げること。自分で勝手に考えると、大間違いをして0点、ということも大いにありうる。だから、数列なども勉強しておくこと。

第1問

次のプログラムは自然数 1 から N までの和を求めるプログラムである。このプログラムについて、次の問いに答えよ。

(配点 20 , ・ A 対応)

```
100 INPUT "n=";N
110 S=0 :A=0
120 FOR I=1 TO N
130     S=S+I
140     IF S<100 THEN GOTO 170
150     PRINT I
160 S=S-100 :A=A+1
170 NEXT I
180 B=S+100*A
190 PRINT "b=";B
200 END
```

- (1) n=?に対して 16 を入力すると、

b=

が表示される。

- (2) n=?に対して 30 を入力すると、

b=

が表示される。

- (3) このプログラムに自然数 k を入力して実行すると、プログラムによって 6 行にわたって、数字や、数字を含む文字列が表示された。このような k の範囲は

k である。

第2問

次のプログラムを実行すると、数の入力を2回求めてくる。
この時、それぞれに自然数を入力するものとして、次の問い
に答えよ。(配点 20, ・B対応)

```
100 INPUT "n=";N
110 INPUT "a=";A
120 S=N : B=0
130 FOR I=2 TO A
140     S=S*N
150 NEXT I
160 GOTO 180
170 S=S/10
180 B=B+1
190 IF S>=10 THEN GOTO 170
200 PRINT B
210 END
```

- (1) $n=?$ に対して5, $a=?$ に対して5を入力すると,

が表示される。
- (2) $n=?$ に対して30, $a=?$ に対してを入力すると,
5が表示される。
- (3) $n=?$ に対して k を入力すると, $a=?$ に対して入力した数と表示される数との差が一定になった。この時, $k =$ であり、その時の, $a=?$ に対して入力した数と表示される数との差はである。
- (4) $n=?$ に対して m を入力すると, $a=?$ に対してどのような自然数を入力しても同じ数が表示された。このとき, $m =$ であり、その時表示される同じ数はである。

練習問題の答え

問題 1 (1)8 (2)32 (3)-57

問題 2 (1)-1 (2)-4

問題 3

(1) $(A-5)^B + (3A-B)$

(2) $\frac{ABC-D}{EF}$

問題 4

$A = \text{SQR}(5) - 2$

$B = (\text{SQR}(5) + 2) / 2 - 2$

問題 5

$\text{SQR}(A * B)$

$A^2 * B^2 + A * B * X - B * C - A * C * X$

問題 6

表を参照

18

問題 7 (1)7 (2)8*5=40

問題 8

表を参照

問題 9

A の 4 乗

問題 10

10 70 80 90 100 70 80 90 100 70 80 90 100 110 20
30 40 50 120 60 130 30 40 50 140

問題 11 3 分の 1

問題 6 の表

行	A	B	C
10	3	5	7
20	6	4	
30	18		8
40			

4 4 9 A=17

B=4

C=9

10

センター試験対策問題の解答(カッコ内は配点)

第 1 問

アイ=14(2)

ウエオ=136(2)

カキ=14(2)

クケ=20(2)

コサ=24(2)

シス=28(2)

セソタ=465(2)

チツ=32(3)

テト=34(3)

第 2 問

ア=4(3)

イ=3(3)

ウエ=10(4)

オ=1(3)

カ=1(3)

キ=1(3)

センター試験対策 コンピュータ

1997 年 12 月 初版発行

1998 年 1 月 改訂版発行

1998 年 9 月 改訂二版発行

1997 ©JTCY